

## Appendix A. HydroDEM Program Source Code

The following Arc Macro Language (AML) program was run to batch-process all the HUCs for this project. The program originally consisted of three separate programs, which were combined after initial processing and testing for a final run of all the data. The program file was named “totalbatch\_mergeNED\_hydrodem\_rerun92.aml”. This

program accomplished the majority of the data processing which could be done in AML. Subsequent steps included running the “Batch Terrain Preprocessing 9” tool in the ArcHydro Tools with the “Catchment Processing Only” option, and building the basin characteristics and global datasets.

```

/* This is a combination of the mergeNED, SetupHydroDEM, and hydrodem amls.
/* run the aml from a top directory.
/* The HUC lists need to be updated for each of the 3 embedded amls if they
/* are to be run.
/* Kenneth Skinner
/* 5/3/06
/* The original amls are kept intact even though some sections have been
/* commented out. A couple routines had to be renumbered.

/* &args huc

/*&if [null %huc%] &then &do
/* &call usage
/* &return
/* &end /* of if-null-huc-then-do
&do huc &list 17040208 17060205 17040209 17060206 17040210 17060207 17040211 17060208 ~
17040212 17060209 17040213 17060210 17040214 17060301 17040215 17060302 17040216 17060303 ~
17040217 17060304 17040218 17060305 17040219 17060306 17040220 17060307 17040207 17060204 ~
16010204 17050104 16020309 17050105 17010101 17050106 17010103 17050107 17010104 17050108 ~
17010105 16010102 17040221 17060308 16010201 17050101 16010202 17050102 16010203 17050103 ~
17050111 17010213 17050112 17010214 17050113 17010215 17050114 17010216 17050115 ~
17010301 17050120 17010302 17050121 17010303 17050122 17010304 17050123 17010305 17050124 ~
17010306 17050201 17010308 17060101 17040104 17060103 17040105 17060106 17040201 17060107 ~
17040202 17060108 17040203 17060109 17040204 17060201 17040205 17060202 17040206 17060203

/* **** LOCAL INSTALLATION CUSTOMIZATIONS below this line ****
&setvar indir D:\hydrodem10
&setvar srcdir D:\hydrodem10\amls /* local directory for source files
&setvar prjfile = %srcdir%\idtm83_newparams.prj /* Default projection file
&sv outdir = D:\hydrodem10\%huc%
&sv huccov = D:\miscproj\ned10m\hucsidtm
&sv tilesc = D:\miscproj\ned10m\tilescov
&sv snapg = D:\miscproj\ned10m\snapg
&sv topdir = D:\miscproj\ned10m
/* **** LOCAL INSTALLATION CUSTOMIZATIONS above this line ****

/* set some default names and settings--these are similar to prototype EDNA
&setvar cellsz = 10 /* NED cell size is 1/3 arc second, approx. 10 meters
&setvar nedmeta = metacov
&setvar metatext = metatext
&setvar tempdem = tempdem
&setvar origdem = dem
&setvar buffdist = 100

```

## 24 Estimated Perennial Streams of Idaho and Related Geospatial Datasets

```
&workspace %indir%
/* Make new workspace, if not already existing
&if ^ [exists %outdir% -directory] &then
    createworkspace %outdir%

/* Clean up grids and coverages from previous runs if they exist
&call cleanup1

/* Select HUC and buffer it
reselect %huccov% %outdir%\huc8 poly
res huc = %huc%
[unquote ``]
n
n

&wo %outdir%
buffer huc8 hucbuff # # %buffdist% 10 poly

grid
verify off
setcell %cellsz%
setwindow hucbuff %topdir%/snapg
buffg = polygrid (hucbuff)
setmask buffg

/* Find the tiles that overlap the buffered HUC poly
reselect %tilesc% poly overlap hucbuff poly
cursor cur declare %tilesc% poly ro
cursor cur open

/* For each selected poly in tiles cover, add path to strg
&sv strg = %topdir%/dem10m/%:cur.tilepath%/demcm
cursor cur next
&do &while %:cur.aml$next% /* while true, still have records to do
    /* build string list of dems
    &sv strg = %strg%, %topdir%/dem10m/%:cur.tilepath%/demcm
    cursor cur next
&end /* of do-while-cur.aml$next (cursor loop through all records)

/* merge tiles to create seamless dem
%origdem% = merge( %strg% )
quit /*leave Grid

/* set PRJs of all data sets
projectdefine grid buffg
&r %prjfile%
projectdefine cover huc8
&r %prjfile%
projectdefine cover hucbuff
&r %prjfile%

&workspace .. /* return to directory in which we started
&call cleanup2

&end /* end of do HUC list
```

```

/* hydrodem.aml -- This aml is used by the National StreamStats Team as the optimal
/* approach for preparing a state's physiographic datasets for watershed delineations.
/* It takes as input, a 10-meter (or 30-foot) DEM, and enforces this data to recognize
/* NHD hydrography as truth. WBD can also be recognized as truth if available for a
/* given state/region. This aml assumes that the DEM has first been projected to a
/* state's projection of choice. This aml prepares data to be used in the Archydro
/* data model (the GIS database environment for National StreamStats).

/* A document describing how to prepare NHD, WBD and dem data for input into this aml
/* can be found on the NHD web site: http://nhd.usgs.gov/watershed/watershed\_tool\_inst\_TOC.html

/* hydrodem.aml supports dem input with either of the following 2 sets of projection parameters:
/* 1. units = meters, cellsize = 10, zunits = 100
/* 2. units = feet, cellsize = 30, zunits = 100
/* hydrodem.aml will reject any other options unless, of course, users modify the aml.
/* zunits is an optional parameter in projection files, but needs to be specified in the input dem
/* before running this aml. zunits represents a value 100 times the 'units' value of the input dataset
/* (if dem input is in meters, the zunits would therefore be centimeters... if dem input is in feet,
/* the zunits values represent 1/100th of a foot. this aml calls agree.aml

&do huc &list 17040208 17060205 17040209 17060206 17040210 17060207 17040211 17060208 ~
17040212 17060209 17040213 17060210 17040214 17060301 17040215 17060302 17040216 17060303 ~
17040217 17060304 17040218 17060305 17040219 17060306 17040220 17060307 17040207 17060204 ~
16010204 17050104 16020309 17050105 17010101 17050106 17010103 17050107 17010104 17050108 ~
17010105 16010102 17040221 17060308 16010201 17050101 16010202 17050102 16010203 17050103 ~
17050111 17010213 17050112 17010214 17050113 17010215 17050114 17010216 17050115 ~
17010301 17050120 17010302 17050121 17010303 17050122 17010304 17050123 17010305 17050124 ~
17010306 17050201 17010308 17060101 17040104 17060103 17040105 17060106 17040201 17060107 ~
17040202 17060108 17040203 17060109 17040204 17060201 17040205 17060202 17040206 17060203

/* **** LOCAL INSTALLATION CUSTOMIZATIONS below this line ****
&sv srcdir = D:\hydrodem10\amls
&sv work = D:\hydrodem10\%huc%
&sv outdir = G:\outdirs\%huc% /* this will be the workspace where all 'local' Archydro
/* workspaces reside.
&sv huc8cov = %work%/huc8 /* the name of the single polygon huc8 coverage
&sv inwall = %work%/inwall /* the name of the coverage of basin boundaries inside the huc8 used for
/* walling this coverage could include gage boundaries and
/* other non-WBD-12-digit boundaries
&sv dendrite = %work%/nhdrch /* the name of the modified nhd reach coverage
&sv origdem = %work%/dem /* the name of the input dem
&sv bowl_polys = %work%/nhd_wbg /* pre-processed NHD water bodies for bowling
&sv bowl_lines = %work%/wb_srcg /* pre-processed NHD centerlines in water bodies
&sv snap_grid = %work%/dem
/* *** REA 5/27/05 added drainplug
&sv drainplug = %work%/drain_plug
/* **** LOCAL INSTALLATION CUSTOMIZATIONS above this line ****

```

## 26 Estimated Perennial Streams of Idaho and Related Geospatial Datasets

```
/* set some default names and settings--these are similar to prototype EDNA and ArchHydro
&setvar cellsz = 10          /* NED cell size is 1/3 arc second, approx. 10 meters
                             /* this variable is changed to 30 below if the units are in feet

&setvar thresh = 150000    /* cell-number threshold for stream initiation and catchments

&setvar thresh2 = 900      /* used to create a stream grid which is later merged with all
                             /* other huc900 grids in the state and converted to shape. This
                             /* shapefile is used in the StreamStats interface for viewing
                             /* and selecting locations on streams.

&setvar filldem = fil      /* final fill grid
&setvar fdirg = fdr       /* final flow direction grid
&setvar faccg = fac       /* final flow accumulation grid

&setvar fsinkg = sinks     /* for QA error check on filled sinks (grid)
&setvar fsinkc = sinks_poly /* for QA error check on filled sinks (poly)

&setvar buffdist = 50      /* value used to buffer the huc8 boundary
&setvar inwallbuffdist = 15 /* value used to buffer the inner walls
&setvar inwallht = 50000   /* inner wall (huc12) height... this variable changed to 150000 (below)
                             /* if units are in feet
&setvar outwallht = 100000 /* outer wall (huc8) height (cm)... this variable changed to 300000
                             /* (below)
                             /* if units are in feet
&setvar bowldepth = 2000   /* The value 2000 (20m) may need to be increased for very wide water
                             /* bodies.

&sv agreebuf = 60         /* agree.aml default value for buffer (when in meters)
&sv agreesmooth = -500    /* agree.aml default value for smooth depth (when in units meters and
                             /* zunits = cm)
&sv agreesharp = -50000   /* agree.aml default value for sharp depth (when in meters)

/* ***** PROJECTION PARAMETER CHECKS BELOW THIS LINE *****
&workspace %work%
&describe %origdem%

&if %PRJ$UNITS% = METERS &then &DO
&if %GRD$DX% = 10 &then &do
&sv cellsz = 10
&sv inwallht = 50000 /* (zunits = cm)
&sv outwallht = 100000 /* (zunits = cm)
&sv agreebuf = 60
&sv agreesmooth = -500 /* (zunits = cm)
&sv agreesharp = -50000 /* (zunits = cm)
&sv buffdist = 50
&sv inwallbuffdist = 15
&end
&else &return if the units are in meters, the cellsize must be 10
&end

&if %PRJ$UNITS% = FEET &then &DO
&if %GRD$DX% = 30 &then &do
&sv cellsz = 30
&sv inwallht = 150000 /* (zunits of 1/100th of a foot)
&sv outwallht = 300000 /* (zunits of 1/100th of a foot)
```

```

&sv agreebuf = 180
&sv agreesmooth = -1500 /* (zunits of 1/100th of a foot)
&sv agreesharp = -150000 /* (zunits of 1/100th of a foot)
&sv buffdist = 150
&sv inwallbuffdist = 45
&end
&else &return if the units are in feet, the cellsize must be 30
&end

&if %PRJ$UNITS% ne FEET and %PRJ$UNITS% ne METERS &then &do
&return units must be in feet or meters
&end

&sv zunits = [quote %PRJ$ZUNITS%]
&if %zunits% ne [quote 100] &then &do
&return zunits must be set and must be 100 times the units of the input dem
&end

/* ***** PROJECTION PARAMETER CHECKS ABOVE THIS LINE *****

/* Clean up grids and coverages from previous runs if they exist
&call cleanup3

&if ^ [exists %outdir% -workspace] &then
    createworkspace %outdir%

&wo %outdir%
buffer %huc8cov% hucbuff # # %buffdist% 1 poly

grid
verify off
setcell %cellsz%
setwindow hucbuff %snap_grid%
buffg = polygrid (hucbuff)
setmask buffg

/* "burning & ridging"

&r %srcdir%/agree.aml %origdem% %dendrite% %agreebuf% %agreesmooth% %agreesharp% /* creates a new
    /* elevation grid -- elevgrid

nhdgrd = linegrid(%dendrite%, #, #, #, #, %cellsz%)

/* Creates the ridged DEM at a height of the DEM + %outwallht%
ridge_n1 = polygrid(%huc8cov%, #, #, #, #, %cellsz%) /* creates a grid of the huc boundaries
ridge_exp = expand(ridge_n1, 2, list, 2) /* expands the present huc grid by 2 cells
/* The next function basically subtracts the previous 2 grids resulting in just the
/* expanded cells representing the wall.
ridge_w = setnull((not isnull(ridge_n1) and not isnull(ridge_exp)), ridge_exp)
dem_ridge8 = elevgrid + con(( not isnull(ridge_w) and isnull(nhdgrd)), %outwallht%, 0)
/* The above statement adds the wall to the original dem except for where streams cross

/* NHD water body 'bowling'
/* NHD streams intersected with water bodies to be used as a source grid.
/* use preexisting water body grid (%bowl_polys%) and source grid (%bowl_lines%).
eucd = setnull(isnull(%bowl_polys%), eucdistance(%bowl_lines%))

```

## 28 Estimated Perennial Streams of Idaho and Related Geospatial Datasets

```
dem_ridge8wb = dem_ridge8 - con( not isnull(eucd), (%bowldepth% / (eucd + 1)), 0)

/* inner wall processing below
&if [exists %inwall% -line] &then &do /* *** REA 6/1/05
  arc buffer %inwall% inwallbuf # # %inwallbuffdist% 1 line
  inwallg_tmp = polygrid(inwallbuf,inside,#,#,%cellsz%)
  inwallg = con(inwallg_tmp == 100,0) /* needed to remove trapped polys inside buffer areas
  /* ("1" values)
  dem_enforced = dem_ridge8wb + con(( not isnull(inwallg) and isnull(nhdgrd)), %inwallht%, 0)
&end /* of if-exists-inwall-then-do /* *** REA 6/1/05
&else /* *** REA 6/1/05
  dem_enforced = dem_ridge8wb /* *** REA 6/1/05
/* end of inner wall processing

/* Do fill, flowdirection, flowaccumulation
setwindow hucbuff %snap_grid%
setcell %origdem%

/* *** REA 5/27 insert drainplug (NODATA holes)
&sv drainplug = %work%/drain_plug
/* If optional drainplug cover given, make its polygons NODATA in DEM before running FILL
&if [exists %drainplug% -polygon] &then &do
  dpg = polygrid (%drainplug%)
  detmp = con(isnull(dpg),dem_enforced)
  kill dem_enforced
  rename detmp dem_enforced
&end /* of if-exists-drainplug-then-do

fill dem_enforced %filldem% sink # %fdirg%
setmask ridge_n1 /* mask back to huc only
%fdirg%msk = %fdirg%
kill %fdirg% all
rename %fdirg%msk %fdirg%

%faccg% = int( flowaccumulation ( %fdirg% ))

/* Identify filled sinks -- look at filled sinks for QA
%fsinkg% = con ( %filldem% gt %origdem%, 1)
%fsinkc% = gridpoly( %fsinkg%)
kill %fsinkg% all

/* Delineate synthetic streams
str = con ( %faccg% ge %thresh%, 1)
lnk = streamlink ( str, %fdirg% )
str900 = con ( %faccg% ge %thresh2%, 1)

/* Delineate catchments from each synthetic reach
cat = watershed (fdr, lnk )

quit /*leave Grid

hillshade dem_enforced shd # # # 0.01
```

```

kill dem_enforced all
kill dem_ridge8 all
kill dem_ridge8wb all
kill elevgrid all
kill eucd all
kill ridge_exp all
kill ridge_nl all
kill ridge_w all

&end /* end of do HUC list

&return /* end of main routine

&routine cleanup3

&do covs &list %fsink% hucbuff inwallbuf
  &if [exists %outdir%/covs% -cover] &then kill %outdir%/covs% all
&end

&do grds &list %fdirg% %faccg% %fsinkg% shd ~
buffg cat dem_ridge8 dem_ridge8wb elevgrid nhdgrd ridge_w ~
str str400 ridge_nl ridge_exp eucd lnk ~
inwallg dem_enforced inwallg_tmp dem_ridge12
  &if [exists %outdir%/grds% -grid] &then kill %outdir%/grds% all
&end

&return /* end of cleanup3 routine

&routine cleanup1

&do infile &list zstat_flowacc
  &if [exists %infile% -info] &then &sv delst [delete %infile% -info]
&end

&do covs &list %nedmeta% huc8 hucbuff
  &if [exists %outdir%/covs% -cover] &then kill %outdir%/covs% all
&end

&do grds &list %tempdem% %origdem% buffg
  &if [exists %outdir%/grds% -grid] &then kill %outdir%/grds% all
&end

&if [exists %outdir%/metatext% -directory] &then &sv delst [delete %outdir%/metatext% -directory]
&return /* end of cleanup1 routine

&routine cleanup2

&do infile &list zstat_flowacc
  &if [exists %infile% -info] &then &sv delst [delete %infile% -info]
&end

&do grds &list %tempdem%
  &if [exists %outdir%/grds% -grid] &then kill %outdir%/grds% all
&end
&return /* end of cleanup2 routine

```

## Appendix B. Program Source Code for Computing Continuous Parameter Grids

The following is an Arc Macro Language (AML) program used to compute continuous parameter grids.

```

/* outdirs2archydrodirs.aml -- Copies relevant datasets from HydroDEM outdirs
/* and computes continuous parameter grids.
/* Run this aml from a top directory.
/* The HUC list may need to be updated for each run, if it does not run to completion.
/*
/* Al Rea 06/06/2008

&do huc &list 17040208 17060205 17040209 17060206 17040210 17060207 17040211 17060208 ~
17040212 17060209 17040213 17060210 17040214 17060301 17040215 17060302 17040216 17060303 ~
17040217 17060304 17040218 17060305 17040219 17060306 17040220 17060307 17040207 17060204 ~
16010204 17050104 16020309 17050105 17010101 17050106 17010103 17050107 17010104 17050108 ~
17010105 16010102 17040221 17060308 16010201 17050101 16010202 17050102 16010203 17050103 ~
17050111 17010213 17050112 17010214 17050113 17010215 17050114 17010216 17050115 ~
17010301 17050120 17010302 17050121 17010303 17050122 17010304 17050123 17010305 17050124 ~
17010306 17050201 17010308 17060101 17040104 17060103 17040105 17060106 17040201 17060107 ~
17040202 17060108 17040203 17060109 17040204 17060201 17040205 17060202 17040206 17060203

/* **** LOCAL INSTALLATION CUSTOMIZATIONS below this line ****
&sv topdir = E:\id_archydro10m
&setvar indir E:\id_archydro10m\outdirs\%huc%
&sv outdir = E:\id_archydro10m\archydro\%huc%
&setvar forestw = E:\id_archydro10m\statewide\forestg
&setvar precipw = E:\id_archydro10m\statewide\precip_prj
/* **** LOCAL INSTALLATION CUSTOMIZATIONS above this line ****

&workspace %indir%

/* Make new workspace, if not already existing
&if ^ [exists %outdir% -directory] &then
    createworkspace %outdir%

/* copy datasets from indir to archydro folders

&do dataset &list cat dem fac fac_mod2 fdr lnk str str900
    &if [exists %indir%\%dataset% -grid] &then copy %indir%\%dataset% %outdir%\%dataset%
&end /* of do-dataset-list

```

```

&do dataset &list %huc%.mdb
  &if [exists %indir%\%dataset% -file] &then &sv cptest [copy %indir%\%dataset% %outdir%\%dataset%]
&end /* of do-dataset-list

/* go to output archydro folder
&wo %outdir%

grid
verify off
setcell dem
setwindow dem dem

/* compute continuous parameter grids for elev, forest, precip
elev_cpg = int(((dem + flowaccumulation(fdr, dem)) / (fac + 1)) * 0.03281 + 0.5)
forest_cpg = int((((%forestw% + flowaccumulation(fdr, %forestw%)) / (fac + 1)) * 100) + 0.5)
precip_cpg = int((((%precipw% + flowaccumulation(fdr, %precipw%)) / (fac + 1)) * 10) + 0.5)

/* compute slope grids
bsldem10m = slope (dem, 0.01, PERCENTRISE)
slop30 = con (bsldem10m ge 30, 1, 0)
aspectg = aspect (dem)
setmask slop30
nfsl30 = con (slop30 gt 0 and (aspectg lt 60 or aspectg gt 300), 1, 0)
setmask off

/* compute continuous parameter grids for mean slope, slopes ge 30, north-facing slopes ge 30
bsld10_cpg = int((bsldem10m + flowaccumulation(fdr, bsldem10m))/(fac + 1)) + 0.5)
slop30_cpg = int((((slop30 + flowaccumulation(fdr, slop30))/(fac + 1)) * 100) + 0.5)
nfsl30_cpg = int((((nfsl30 + flowaccumulation(fdr, nfsl30))/(fac + 1)) * 100) + 0.5)

/* clean up
kill aspectg all

quit /* out of grid

&workspace %topdir%

&end /* end of do HUC list

```